query: #defaultView:Map{"hide":"?rgb"}
  SELECT ?item ?itemLabel ?itemDescription (GROUP_CONCAT(DISTINCT ?whereLabel; SEPARATOR=', ') AS ?whereLabels) (SAMPLE(?image) AS ?image) ?coord ?rgb ?layer WITH {
    SELECT DISTINCT ?item WHERE {
      ?item wdt:P17|wdt:P131* wd:Q665641 .
    }
  } AS %subquery1 WHERE {
    INCLUDE %subquery1 .
    ?item wdt:P31 [] .
    ?item p:P625 ?coordStatement .
    ?coordStatement ps:P625 ?coord .
    #MINUS { ?coordStatement prov:wasDerivedFrom/pr:P143 wd:Q169514 } # imported from Wikimedia project: Swedish Wikipedia 
    #MINUS { ?coordStatement prov:wasDerivedFrom/pr:P143 wd:Q837615 } # imported from Wikimedia project: Cebuano Wikipedia 
    #MINUS { ?coordStatement prov:wasDerivedFrom/pr:P248 wd:Q1194038 } # stated in: GEOnet Names Server
    OPTIONAL {
      ?item wdt:P131 ?where .
      OPTIONAL {
        ?where rdfs:label ?whereLiteral .
        FILTER(LANG(?whereLiteral) = 'de') .
      }
    }
    BIND(IF(BOUND(?where), IF(BOUND(?whereLiteral), ?whereLiteral, STRAFTER(STR(?where), 'entity/')), 'no P131') AS ?whereLabel) .
    OPTIONAL { ?item wdt:P18 ?image }
    BIND(IF(EXISTS { ?item p:P18 [] }, '0000ff', 'ff0000') AS ?rgb) .
    BIND(IF(EXISTS { ?item p:P18 [] }, 'With image', 'Without image') AS ?layer) .
    SERVICE wikibase:label { bd:serviceParam wikibase:language '[AUTO_LANGUAGE,de,en]' }
  } GROUP BY ?item ?itemLabel ?itemDescription ?whereLabels ?coord ?rgb ?layer ORDER BY ASC(?item)